F04AFF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F04AFF calculates the accurate solution of a set of real symmetric positive-definite linear equations with multiple right-hand sides, $AX = B$, with iterative refinement, where A has been factorized by F03AEF.

2 Specification

```
 SUBROUTINE F04AFF(N, IR, A, IA, P, B, IB, EPS, X, IX, BB, IBB, K,
1                  IFAIL)
 INTEGER           N, IR, IA, IB, IX, IBB, K, IFAIL
 real              A(IA,N), P(N), B(IB,IR), EPS, X(IX,IR),
1                  BB(IBB,IR)
```

3 Description

To solve a set of real linear equations $AX = B$ where A is symmetric positive-definite, the routine must be preceded by a call to F03AEF which computes a Cholesky factorization of A as $A = LL^T$, where L is lower triangular. An approximation to X is then found by forward and backward substitution. The residual matrix $R = B - AX$ is then calculated using ***additional precision***, and a correction D to X is found by solving $LL^TD = R$. X is replaced by $X + D$, and this iterative refinement of the solution is repeated until full machine accuracy has been obtained.

4 References

[1] Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

5 Parameters

1: N — INTEGER *Input*

On entry: n, the order of the matrix A.

2: IR — INTEGER *Input*

On entry: r, the number of right-hand sides.

3: A(IA,N) — ***real*** array *Input*

On entry: the upper triangle of the n by n positive-definite symmetric matrix A, and the sub-diagonal elements of its Cholesky factor L, as returned by F03AEF.

4: IA — INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04AFF is called.

Constraint: IA $\geq$ N.

5: P(N) — ***real*** array *Input*

On entry: the reciprocals of the diagonal elements of L, as returned by F03AEF.

6: B(IB,IR) — ***real*** array *Input*

On entry: the n by r right-hand side matrix B.

7: IB — INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F04AFF is called.

Constraint: IB $\geq$ N.

8: EPS — ***real*** *Input*

On entry: EPS must be set to the value of the ***machine precision***.

9: X(IX,IR) — ***real*** array *Output*

On exit: the n by r solution matrix X.

10: IX — INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which F04AFF is called.

Constraint: IX $\geq$ N.

11: BB(IBB,IR) — ***real*** array *Output*

On exit: the final n by r residual matrix $R = B - AX$.

12: IBB — INTEGER *Input*

On entry: the first dimension of the array BB as declared in the (sub)program from which F04AFF is called.

Constraint: IBB $\geq$ N.

13: K — INTEGER *Output*

On exit: the number of iterations needed in the refinement process.

14: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

The matrix A is too ill-conditioned to produce a correctly rounded solution.

7 Accuracy

The computed solutions should be correct to full machine accuracy. For a detailed error analysis see Wilkinson and Reinsch [1] page 39.

8 Further Comments

The time taken by the routine is approximately proportional to $n^2 r$.

9 Example

To solve the set of linear equations $AX = B$ where

$$A = \begin{pmatrix} 5 & 7 & 6 & 5 \\ 7 & 10 & 8 & 7 \\ 6 & 8 & 10 & 9 \\ 5 & 7 & 9 & 10 \end{pmatrix} \text{ and } B = \begin{pmatrix} 23 \\ 32 \\ 33 \\ 31 \end{pmatrix}.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F04AFF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NMAX, IR, IA, IB, IX, IBB
      PARAMETER        (NMAX=8,IR=1,IA=NMAX,IB=NMAX,IX=NMAX,IBB=NMAX)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             D1, EPS
      INTEGER          I, ID, IFAIL, J, K, N
*     .. Local Arrays ..
      real             A(IA,NMAX), B(IB,IR), BB(IBB,IR), P(NMAX),
     +                 X(IX,IR)
*     .. External Functions ..
      real             X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         F03AEF, F04AFF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F04AFF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GT.0 .AND. N.LE.NMAX) THEN
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
         IFAIL = 1
*
*        Cholesky decomposition
         CALL F03AEF(N,A,IA,P,D1,ID,IFAIL)
*
         IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99999) 'Error in F03AEF. IFAIL =', IFAIL
         ELSE
            READ (NIN,*) ((B(I,J),J=1,IR),I=1,N)
            EPS = X02AJF()
            IFAIL = 1
*
*           Accurate solution of linear equations
            CALL F04AFF(N,IR,A,IA,P,B,IB,EPS,X,IX,BB,IBB,K,IFAIL)
*
            IF (IFAIL.NE.0) THEN
               WRITE (NOUT,99999) 'Error in F04AFF. IFAIL =', IFAIL
            ELSE
               WRITE (NOUT,*) ' Solution'
```

```
                DO 20 I = 1, N
                   WRITE (NOUT,99998) (X(I,J),J=1,IR)
   20           CONTINUE
             END IF
          END IF
       ELSE
          WRITE (NOUT,99999) 'N is out of range: N = ', N
       END IF
       STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,8F9.4)
      END
```

9.2 Program Data

```
F04AFF Example Program Data
  4
    5    7    6    5
    7   10    8    7
    6    8   10    9
    5    7    9   10
   23   32   33   31
```

9.3 Program Results

```
 F04AFF Example Program Results

  Solution
    1.0000
    1.0000
    1.0000
    1.0000
```